Parabola Problems

#1 Center of Mass

Parabola has equation $y = f(x) = x^2$. Let R_h be the convex region defined by

$$R_h = \left\{(x, y) \in \mathbb{R}^2 : f(x) \le y \le h\right\}$$

whenever $h \ge 0$.

If we assume R_h has a uniform area density $\sigma = 1$, the center of mass is given by:

In[1]:=
```
ρ[h_] := 2 √y;
comY[h_] := Integrate[y ρ[y], {y, 0, h}] / Integrate[ρ[y], {y, 0, h}];
```

In[3]:=
```
comY[h]
```

Out[3]= $\frac{3\,h}{5}$

So the center of mass is at $(0, 3\,h/5)$.

#2 Stable Equilibrium

The equilibrium is stable if rolling slightly to either side increases the height of the center of mass.
If we fix a frame of reference relative to the “ground” we can see the height of the COM is measured as the perpendicular distance from the COM to the “ground” line which is always tangent to the parabola as it rolls.
We can instead focus on a frame fixed relative to the parabola itself and consider the perpendicular distance from the COM to various tangent lines. Specifically let L_x be a line tangent to R_h at the point $\left(x, x^2\right)$. We proceed to find the distance from the COM to this line. In physics this is essentially the potential energy of the parabola so well denote it by $U(x)$. A stable equilibrium occurs where $U(x)$ has a local minimum.

The line contains the point $p = (x, f(x))$ and extends in the direction of the vector $v = (1, f'(x))$.
The vector $w = \langle f'(x), -1\rangle$ is perpendicular to it.
So the distance to the center of mass c is given by:

$$U(x) = \left\|\mathrm{proj}_w(c - p)\right\| = (p - c)\cdot\left(\frac{w}{\|w\|}\right)$$

In[81]:=
```
c = {0, 3 h / 5};
p = {x, x^2};
v = {1, 2 x};
w = {2 x, -1};
```

In[85]:=
```
Simplify[(p - c).w / Norm[w], x ∈ Reals]
```

Out[85]= $\frac{\frac{3\,h}{5} + x^2}{\sqrt{1 + 4\,x^2}}$

The distance is

$$U(x) = \frac{\frac{3h}{5} + x^2}{\sqrt{1 + 4x^2}}$$

In[86]:=
```
ClearAll[u];
u[x_, h_] := (3 h/5 + x^2)/Sqrt[1 + 4 x^2];
```

In[177]:=
```
Manipulate[Show[{
   Graphics[{Red, Point[{0, 3 h / 5}],
     Gray, Dashed, Line[{{-Sqrt[h], h}, {Sqrt[h], h}}]}],
   Plot[Evaluate@{
      x^2,
      u[x, h],
      D[u[x, h], {x, 2}]}, {x, -3, 7},
    PlotLegends -> {"f(x)", "u", "du^2/dx^2"}]
  },
  PlotRange -> {{-2, 3}, {-0.5, Max[2 h, 3]}},
  Axes -> True], {{h, 5 / 6}, 0, 10}]
```

Out[177]=



U has a local minimum at 0 when $U''(0) \geq 0$ (and we already know that $U'(x) = 0$).
The threshold height h is given by:

In[90]:=
```
Solve[(x ↦ u[x, h])''[0] == 0, h]
```

Out[90]= $\left\{\left\{h \to \frac{5}{6}\right\}\right\}$

At this height the center of mass is at (0, 1/2).

#3 Conditions for 45° Equilibrium

In the parabola's frame, resting at 45° amounts to the "ground" being a tangent line with a slope of 1.
The tangent direction $(1, 2x)$ is equal to $(1, 1)$ when $x = 1/2$. If this rotation provides a stable equilibrium then we must have $U'(1/2) = 0$ and $U''(1/2) \geq 0$.

In[98]:=
```
Solve[(x ↦ u[x, h])'[1 / 2] == 0, h]
```

Out[98]= $\left\{\left\{h \to \frac{5}{4}\right\}\right\}$

This is our only candidate solution. Two properties must be checked:
1) When the parabola is limited to the height 5/4 the right-most point has an x coordinate of $\sqrt{h} \approx 1.118 > 1/2$ so the point of tangency is part of the parabolic section.
2) $U''(1/2) = \frac{1}{\sqrt{2}} > 0$ when $h = 5/4$ so $x = 1/2$ is a local minimum for U.

#4 Conditions for rolling 1 unit

We address two properties one at a time:
1) If the point of tangency (in a "lab" frame) is at (1, 0) then in the parabola's frame the point is an arclength-distance of 1 away from the vertex.
2) If the parabola is at rest with this point of tangency then the point corresponds to a local minimum of the potential.

First, the arc length:

In[109]:=
```
s[a_] := Integrate[√(1 + (2 x)^2), {x, 0, a}];
Simplify[s[x], x > 0]
```

Out[110]= $\frac{1}{4}\left(2x\sqrt{4x^2+1} + \sinh^{-1}(2x)\right)$

If $s(x) = 1$ then

$$\sinh^{-1}(2x) == 4 - 2x\sqrt{4x^2+1}$$

Given $\sinh(\alpha) = \frac{1}{2}e^{\alpha} - \frac{1}{2}e^{-\alpha}$ we take sinh of both sides to get:

$$2x = \frac{1}{2}\left(\frac{e^4}{e^{2x\sqrt{4x^2+1}}} - \frac{e^{2x\sqrt{4x^2+1}}}{e^4}\right)$$

so

$$4x = \frac{e^4}{e^{2x\sqrt{4x^2+1}}} - \frac{e^{2x\sqrt{4x^2+1}}}{e^4} = B - \frac{1}{B}$$

By solving $4x = B - 1/B$ for B we get

$$B = 2x \pm \sqrt{1+4x^2}$$

so we can then solve

$$\frac{e^4}{e^{2x\sqrt{4x^2+1}}} = B = 2x +^{(1)} \sqrt{1+4x^2}$$

Since $B > 0$ we must use the + solution.

We get a solution for x which internally is the root of a complicated expression, which displays as a numerical approximation of the exact quantity:

In[162]:=
```
x$solution = Solve[ E^4 / E^(2 x Sqrt[4 x^2+1]) == 2 x + Sqrt[1 + 4 x^2] && x > 0, {x}]
xsol = N[First[x /. x$solution], 16]
```

Out[162]= {{x → 0.764...}}

Out[163]= 0.7639266633170910

Second, we assert $U'(x) = 0$ and solve for h with this constraint:

In[154]:=
```
h$solution = Solve[(x ↦ u[x, h])'[xsol] == 0, h]
hsol = First[h /. h$solution]
```

Out[154]= {{h → 1.80597324487797}}

Out[155]= 1.80597324487797

Finally we check that for this value of h, $U'(x) = 0$ and $U''(x) > 0$:

In[158]:=
```
D[u[x, h], {x, 1}] /. {x → xsol, h → hsol}
D[u[x, h], {x, 2}] /. {x → xsol, h → hsol}
```

Out[158]= $0. \cdot 10^{-16}$

Out[159]= 0.76679508993478

#5 BONUS PROBLEM

If the parabola starts out balanced vertically ($h > 5/6$) and starts rolling to the right without slipping,
what height must the parabola be to continue rolling so far that it pivots on the boundary corner, continues turning over, and falls flat against the ground? In order to have enough energy to pivot over its corner, it has to start out sufficiently tall.

Hint: The logic is a few steps if you've seen conservation of kinetic / potential energy in physics
and the property you need can be phrased in a super elegant way.
The correct answer h^* should make the quantity $2\cos(\pi/h^*)$ a familiar number.